Software System Design and Implementation

GADTs Practice

Curtis Millar
CSE, UNSW
23 July 2021

Exercise 5

- Parse a series of tokens.
- Stack push and pop.
- Evaluate a sequence of tokens.
- Calculate a string.

Recall: GADTs

Generalised Algebraic Datatypes (*GADTs*) is an extension to Haskell that, among other things, allows data types to be specified by writing the types of their constructors:

```
{-# LANGUAGE GADTs, KindSignatures #-}
-- Unary natural numbers, e.g. 3 is S (S (S Z))
data Nat = Z | S Nat
-- is the same as
data Nat :: * where
  Z :: Nat
  S :: Nat -> Nat
```

The `printf` function

Consider the well known C function `printf`:

```
printf("Hello %s You are %d years old!", "Nina", 22)
```

In C, the type (and number) of parameters passed to this function are dependent on the first parameter (the format string).

The printf function

To do a similar thing in Haskell, we would like to use a richer type that allows us to define a function whose subsequent parameter is determined by the first.

```
{-# LANGUAGE StandaloneDeriving #-}
data Format :: * -> * where
  End :: Format ()                          -- Empty format
  Str :: Format t -> Format (String, t)     -- %s
  Dec :: Format t -> Format (Int, t)        -- %d
  L :: String -> Format t -> Format t   -- literal strings
-- just like deriving (Show) for normal data types.
deriving instance Show (Format ts)
```

Our format strings are indexed by a tuple type containing all of the types of the %directives used.

The printf function

```
"Hello %s You are %d years old!"
```

is written:

```
L "Hello" $ Str $ L " You are "
     $ Dec $ L " years old!" End
```

The printf function

```
printf :: Format ts -> ts -> IO ()
printf End () =
    pure () -- type is ()
printf (Str fmt) (s,ts) =
    do putStr s; printf fmt ts -- type is (String, ...)
printf (Dec fmt) (i,ts) =
    do putStr (show i); printf fmt ts -- type is (Int,..)
printf (L s fmt) ts     =
    do putStr s; printf fmt ts
```

Vectors

Define a natural number kind to use on the type level:

```
data Nat = Z | S Nat
```

Our length-indexed list can be defined, called a `Vec`:

```
{-# LANGUAGE DataKinds #-}
data Vec (a :: *) :: Nat -> * where
  Nil  :: Vec a Z
  Cons :: a -> Vec a n -> Vec a (S n)
```

The functions `hd` and `tl` can be total:

```
hd :: Vec a (S n) -> a
hd (Cons x xs) = x
tl :: Vec a (S n) -> Vec a n
tl (Cons x xs) = xs
```

Vectors, continued

Our map for vectors is as follows:

```
mapVec :: (a -> b) -> Vec a n -> Vec b n
mapVec f Nil = Nil
mapVec f (Cons x xs) = Cons (f x) (mapVec f xs)
```

Properties

Using this type, it's impossible to write a mapVec function that changes the length of the vector.

Properties are verified by the compiler!

Appending Vectors

Example (Problem)

```
appendV :: Vec a m -> Vec a n -> Vec a ???
```

We want to write `m + n` in the `???` above, but we do not have addition defined for kind `Nat`.

We can define a normal Haskell function easily enough:

```
plus :: Nat -> Nat -> Nat
plus Z y = y
plus (S x) y = S (plus x y)
```

This function is not applicable to type-level `Nat`s, though.

$\Rightarrow$ we need a type level function.

Type Families

Type level functions, also called *type families*, are defined in Haskell like so:

```
{-# LANGUAGE TypeFamilies #-}
type family Plus (x :: Nat) (y :: Nat) :: Nat where
  Plus Z     y = y
  Plus (S x) y = S (Plus x y)
```

We can use our type family to define appendV:

```
appendV :: Vec a m -> Vec a n -> Vec a (Plus m n)
appendV Nil         ys = ys
appendV (Cons x xs) ys = Cons x (appendV xs ys)
```

Concatenating Vectors

Example (Problem)

```
concatV :: Vec (Vec a m) n -> Vec a ???
```

We want to write `m * n` in the `???` above, but we do not have times defined for kind `Nat`.

```
{-# LANGUAGE TypeFamilies #-}
type family Times (a :: Nat) (b :: Nat) :: Nat where
  Times Z     m = Z
  Times (S n) m = Plus m (Times n m)
```

We can use our type family to define `concatV`:

```
{-# LANGUAGE UndecidableInstances #-}
concatV :: Vec (Vec a n) m -> Vec a (Times n m)
concatV Nil         = Nil
concatV (Cons v vs) = v `appendV` concatV vs
```

Filtering Vectors

Example (Problem)

```
filterV :: (a -> Bool) -> Vec a n -> Vec a ???
```

What is the size of the result of filter?

Filtering Vectors

Example (Problem)

```
filterV :: (a -> Bool) -> Vec a n -> [a]
```

We do not know the size of the result.

We can use our type family to define `concatV`:

```
filterV :: (a -> Bool) -> Vec a n -> [a]
filterV p Nil = []
filterV p (Cons x xs)
                | p x  = x : filterV p xs
                | otherwise = filterV p xs
```

Homework

1. Assignment 2 is **released**. The deadline has been extended; it is now due on **Wednesday 4th August, 6 PM** (in 12 days).
2. Week 7's quiz is due on today. Make sure you submit your answers.
3. The sixth programming exercise is due by the start of my next lecture (in 7 days).
4. This week's quiz is also up, it's due Friday next week at 6pm (in 7 days).

Consultations

- Consultations will be made on request. Ask on course forum or email `cs3141@cse.unsw.edu.au`.
- If there is a consultation it will be announced on the course forum with a link a room number for Hopper.
- Make sure to join the queue on Hopper. Be ready to share your screen with REPL (`ghci` or `stack repl`) and editor set up.